Maple-assisted proof of empirical formula for A203884

There are $7^6 = 117649$ possible rows, but instead of defining a 117649×117649 matrix we can take advantage of symmetry to work with "states" corresponding to the $A000110(6) = 203$ partitions of $\{1,\ldots,6\}$ into subsets. Given a state s, let A_s be the set of assignments x of numbers $0 \ldots 6$ to $1 \ldots 6$ so that $x_i = x_j$ if and only if i and j are in the same set of the partition. Given states s and t and any assignment $x \in A_s$, let T_{st} be the number of $y \in A_t$ such that x and y could be used in adjacent rows of an array counted by this sequence, i.e. for i from 1 to 5, $x_i = y_{i+1}$ or $y_i = x_{i+1}$.

```
> States:= combinat:-setpartition([$1..6]):
  T:= Matrix(203,203):
  for ii from 1 to 203 do
    s:= States[ii];
    X:= Vector(6):
    for i from 1 to nops(s) do X[s[i]]:= i od:
    for jj from 1 to 203 do
      t:= States[jj];
      nt:= nops(t);
      Y:= Vector(6);
      for p in combinat:-permute([$0..6],nt) do
         for i from 1 to nt do Y[t[i]]:= p[i] od:
         good:= true;
         for i from 1 to 5 do if X[i]<>Y[i+1] and Y[i]<>X[i+1] then
  good:= false; break fi od;
         if good then T[ii,jj]:= T[ii,jj]+1 fi
      od
    od;
  od:
```

Now we should have $a(n) = u^T T^n v$ where $u_i = |A_{s(i)}|$ and $t_i = 1$ for all i. To verify this, I will compute the first few terms of the sequence. It will be useful to compute $T^n v$ iteratively rather than actually taking powers of T.

```
> u:= Vector[row](203, i -> 7!/(7-nops(States[i]))!):
  v:= Vector(203,1):
  Tv[0]:= v:
  for nn from 1 to 12 do Tv[nn]:= T . Tv[nn-1] od:
> seq(u . Tv[nn], nn=1..10);
```

18193357, 6601376089, 2880436707253, 1337957259758497, 632382967906147549, 300614697430789927465, 143131084226139999444805, 68183470583835920750899441, 32485295368125261340573143277, 15477975279195312357768613147129 **(1)**

Now the empirical formula is

```
> Emp:= a(n) = 565*a(n-1) -26916*a(n-2) -8476092*a(n-3) +587809224*
  a(n-4) -6796290096*a(n-5) -158295697632*a(n-6) +2605874815296*a
  (n-7) -2681418764160*a(n-8) -55173588410880*a(n-9)
  +65536012915200*a(n-10) +108491436288000*a(n-11)
  -105573943296000*a(n-12)
```

$$Emp := a(n) = 565\, a(n-1) - 26916\, a(n-2) - 8476092\, a(n-3) + 587809224\, a(n-4) - 6796290096\, a(n-5) - 158295697632\, a(n-6) + 2605874815296\, a(n-7) - 2681418764160\, a(n-8) - 55173588410880\, a(n-9) + 65536012915200\, a(n-10) \quad (2)$$

$+ 108491436288000\, a(n-11) - 105573943296000\, a(n-12)$

This corresponds to saying $u^T\, T^n P(T)\, v = 0$ where P is the following polynomial.

```
> P:= x^12 - add(coeff(rhs(Emp),a(n-j))*x^(12-j),j=0..12);
```

$$P := x^{12} - 565\, x^{11} + 26916\, x^{10} + 8476092\, x^9 - 587809224\, x^8 + 6796290096\, x^7 + 158295697632\, x^6 - 2605874815296\, x^5 + 2681418764160\, x^4 + 55173588410880\, x^3 - 65536012915200\, x^2 - 108491436288000\, x + 105573943296000 \quad (3)$$

We compute $P(T)\, v$ using the previously computed values of $T^n v$, and verify that it is 0 :

```
> Q:= add(coeff(P,x,j)*Tv[j],j=0..12):
  LinearAlgebra:-Equal(Q,Vector(203));
```

true **(4)**

Thus we have $u^T\, T^n P(T)\, v = 0$. This completes the proof.